

V2X 스마트 지팡이 — 4차 집중 미팅(5시간+) 팀원별 단계형 실행 계획서

작성 목적: 단순 설명서가 아니라, 미팅 당일 각 팀원이 자기 파트만 보고 순서대로 실행할 수 있는 작업 지시서로 구성한다.

4차 미팅 핵심 산출물: 실제 GPS·IMU 센서 데이터 ESP-NOW 전송, 수신기 ESP32의 Jetson Nano UART 전달, 진동·음성 액추에이터 제어, SUMO FCD 궤적 데이터 추출.

0. 오늘의 전체 그림

0.1 지난 시간까지 완료한 것

담당자	완료 항목	의미	4차에서 이어질 작업
박중선	GPS, IMU 센서 ESP32 연결 및 데이터 출력 성공	센서값을 실제로 읽을 수 있음	센서값을 ESP-NOW 구조체에 매핑하여 송신
강현준	ESP32 간 ESP-NOW 1:1 통신 80% 성공	더미 데이터 무선 전송 가능	실제 센서 패킷 수신 후 Jetson으로 UART 전달
최민서	SUMO OS 충돌 해결 및 주차장 진출입로 맵 Import 완료	AI 학습용 시뮬레이션 기반 확보	차량 Flow 주입, FCD XML 생성, CSV 전처리

0.2 오늘 각자 해야 할 핵심 한 줄 요약

담당자	오늘의 미션	완료 기준
박중선	송신기 ESP32에 GPS+IMU를 연결하고 실제 센서값을 ESP-NOW로 송신하며, 수신기 측 진동/음성 회로를 테스트한다.	송신기 보드를 흔들거나 이동하면 수신기에서 가속도·GPS·risk_level이 바뀌고, risk_level에 따라 진동/음성이 동작한다.
강현준	수신기 ESP32 코드를 구조체 기준으로 정리하고 Jetson Nano와 UART 파이프라인을 완성한다.	Jetson 터미널에서 seq, lat, lng, accel, risk가 실시간으로 파싱되어 출력되고 CSV 로그가 저장된다.
최민서	SUMO 맵에 다양한 차량 Flow를 주입하고 FCD 궤적 데이터를 Python으로 전처리한다.	sumo-gui에서 차량 20대 이상이 움직이고 fcd_trajectory.csv가 생성된다.

0.3 오늘 사용할 공통 데이터 구조

중요: 송신기와 수신기 코드의 v2x_message_t 구조체는 1바이트도 달라지면 안 된다. 변수 순서, 자료형, packed 속성을 그대로 복사해서 사용한다.

필드	자료형	설명	예시
node_type	uint8_t	노드 종류	0x01 = 지팡이

risk_level	uint8_t	위험도	0 안전, 1 주의, 2 경고, 3 위험
gps_valid	uint8_t	GPS 좌표 유효 여부	실내 0, 야외 수신 후 1
latitude, longitude	float	GPS 좌표	37.XXXXXX, 126.XXXXXX
speed_mps, heading_deg	float	속도, 방향각	m/s, degree
accel_x/y/z	float	가속도	m/s ²
gyro_x/y/z	float	각속도	dps
timestamp_ms	uint32_t	송신 시각	millis()
seq_num	uint16_t	패킷 번호	0, 1, 2, 3...

파트 A. 박종선 — 송신기 H/W 통합 + 실제 센서 ESP-NOW 송신

오늘의 목표: GPS와 IMU가 연결된 송신기 ESP32에서 실제 센서값을 읽고, 강현준의 수신기 ESP32로 100ms 간격 ESP-NOW 패킷을 보낸다.

A-1. 준비물 체크 0~10분

체크	준비물	개수	확인 방법
☐	ESP32 DevKit 송신기	1개	보드에 “송신기” 라벨 부착
☐	ICM-20948 IMU 모듈	1개	I2C 주소 0x69 또는 0x68 확인
☐	Neo-6M GPS 모듈	1개	안테나 방향 확인
☐	브레드보드/점퍼선	충분히	헐거운 선은 즉시 교체
☐	USB 케이블	1개	데이터 전송 가능한 케이블
☐	Arduino IDE	1개	ESP32 보드 매니저 설치

절대 규칙: 배선 중에는 USB 전원을 켜지 않는다. 전원이 들어간 상태에서 VCC/GND가 순간적으로 닿으면 ESP32, GPS, IMU가 손상될 수 있다.

A-2. 송신기 회로 배선 10~25분

A-2-1. 전원 레일 구성

출발	도착	색상 권장	의미
ESP32 3V3	브레드보드 + 레일	빨간색	센서 3.3V 전원
ESP32 GND	브레드보드 - 레일	검은색	공통 접지

A-2-2. ICM-20948 IMU 연결

ICM-20948	ESP32	통신	확인 포인트
VCC	3V3 레일	전원	5V 금지, 모듈 사양 확인
GND	GND 레일	접지	공통 GND
SDA	GPIO 21	I2C 데이터	SCL과 바뀌면 인식 안 됨
SCL	GPIO 22	I2C 클럭	기본 I2C 핀

A-2-3. Neo-6M GPS 연결

Neo-6M	ESP32	통신	확인 포인트
VCC	3V3 또는 모듈 허용 5V	전원	모듈 보드에 레귤레이터가 있으면 5V 가능
GND	GND 레일	접지	공통 GND

TX	GPIO 16 RX2	GPS → ESP32	TX와 RX는 교차
RX	GPIO 17 TX2	ESP32 → GPS	수신만 해도 되지만 연결 권장

사용 금지 핀: GPIO 6~11은 내부 플래시 연결 핀이라 사용 금지. GPIO 1/3은 USB Serial, GPIO 0/2/5/12/15는 부팅 스트래핑 핀이므로 센서 연결을 피한다.

A-3. 라이브러리 설치 25~35분

1. Arduino IDE → Library Manager를 연다.
2. **TinyGPSPlus**를 검색하여 설치한다.
3. **SparkFun ICM-20948 Arduino Library**를 검색하여 설치한다.
4. ESP32 보드 패키지가 설치되어 있는지 확인한다.

A-4. IMU I2C 스캐너 테스트 35~45분

```
#include <Wire.h>

#define I2C_SDA 21
#define I2C_SCL 22

void setup() {
  Serial.begin(115200);
  delay(1000);
  Wire.begin(I2C_SDA, I2C_SCL);
  Serial.println("=== I2C Scanner Start ===");
}

void loop() {
  int count = 0;
  for (byte addr = 1; addr < 127; addr++) {
    Wire.beginTransmission(addr);
    if (Wire.endTransmission() == 0) {
      Serial.print("I2C device found: 0x");
      if (addr < 16) Serial.print("0");
      Serial.println(addr, HEX);
      count++;
    }
  }
  if (count == 0) Serial.println("No I2C device. Check VCC/GND/SDA/SCL.");
  else {
    Serial.print("Total device count: ");
    Serial.println(count);
  }
  Serial.println("-----");
  delay(3000);
}
```

기대 출력: I2C device found: 0x69 또는 0x68. 아무것도 안 뜨면 VCC/GND/SDA/SCL 순서로 다시 확인한다.

A-5. GPS 단독 테스트 45~60분

```
#include <TinyGPSPlus.h>

#define GPS_RX 16
#define GPS_TX 17
#define GPS_BAUD 9600

TinyGPSPlus gps;
HardwareSerial gpsSerial(2);

void setup() {
  Serial.begin(115200);
  gpsSerial.begin(GPS_BAUD, SERIAL_8N1, GPS_RX, GPS_TX);
  Serial.println("=== GPS Raw/Parsed Test ===");
}

void loop() {
  while (gpsSerial.available() > 0) gps.encode(gpsSerial.read());

  if (gps.location.isUpdated()) {
    Serial.print("lat=");
    Serial.print(gps.location.lat(), 6);
    Serial.print(" lng=");
    Serial.print(gps.location.lng(), 6);
  }
}
```

```

    Serial.print(" satellites=");
    Serial.print(gps.satellites.value());
    Serial.print(" speed_mps=");
    Serial.print(gps.speed.mps(), 2);
    Serial.print(" course_deg=");
    Serial.println(gps.course.deg(), 2);
}

static uint32_t last = 0;
if (millis() - last > 3000) {
    last = millis();
    Serial.print("chars=");
    Serial.print(gps.charsProcessed());
    Serial.print(" valid=");
    Serial.println(gps.location.isValid());
}
}
}

```

GPS 주의: 실내에서는 위도/경도가 안 잡히는 것이 정상이다. 야외 또는 창가에서 1~3분 Cold Start를 기다린다. gps.charsProcessed()가 증가하면 UART 배선은 살아 있는 것이다.

A-6. 수신기 MAC 주소 반영 60~65분

1. 강현준이 B-2에서 출력한 수신기 MAC 주소를 받는다.
2. 예: C8:F0:9E:52:78:A4 → {0xC8, 0xF0, 0x9E, 0x52, 0x78, 0xA4} 형태로 바꾼다.
3. 아래 Sender 코드의 **receiverMAC[]**에 정확히 입력한다.

A-7. 송신기 최종 코드 업로드 65~95분

```

#include <WiFi.h>
#include <esp_now.h>
#include <Wire.h>
#include <TinyGPSPlus.h>
#include <ICM_20948.h>

#define NODE_CANE 0x01
#define RISK_SAFE 0
#define RISK_CAUTION 1
#define RISK_WARNING 2
#define RISK_DANGER 3

#define I2C_SDA 21
#define I2C_SCL 22
#define GPS_RX 16
#define GPS_TX 17
#define GPS_BAUD 9600
#define SEND_PERIOD_MS 100
#define AD0_VAL 1

uint8_t receiverMAC[] = {0xC8, 0xF0, 0x9E, 0x52, 0x78, 0xA4};

typedef struct __attribute__((packed)) v2x_message {
    uint8_t node_type;
    uint8_t risk_level;
    uint8_t gps_valid;
    float latitude;
    float longitude;
    float speed_mps;
    float heading_deg;
    float accel_x;
    float accel_y;
    float accel_z;
    float gyro_x;
    float gyro_y;
    float gyro_z;
    uint32_t timestamp_ms;
    uint16_t seq_num;
} v2x_message_t;

ICM_20948 I2C imu;
TinyGPSPlus gps;
HardwareSerial gpsSerial(2);
v2x_message_t packet;
uint16_t seq = 0;
uint32_t lastSendMs = 0;
float lastLat = 0.0;
float lastLng = 0.0;
float lastSpeed = 0.0;
float lastHeading = 0.0;
uint8_t lastGpsValid = 0;

void onSent(const uint8_t *mac, esp_now_send_status_t status) {
    Serial.print("[ESP-NOW TX] seq=");
    Serial.print(seq - 1);
    Serial.println(status == ESP_NOW_SEND_SUCCESS ? " result=OK" : " result=FAIL");
}

```

```

}

void setupImu() {
    Wire.begin(I2C_SDA, I2C_SCL);
    Wire.setClock(400000);
    imu.begin(Wire, AD0_VAL);
    if (imu.status == ICM_20948_Stat_OK) {
        Serial.println("[IMU] connected. expected address: 0x69");
    } else {
        Serial.print("[IMU] failed. status=");
        Serial.println(imu.statusString());
    }
}

void setupGps() {
    gpsSerial.begin(GPS_BAUD, SERIAL_8N1, GPS_RX, GPS_TX);
    Serial.println("[GPS] UART2 started. GPS_TX -> ESP32_RX16, GPS_RX -> ESP32_TX17");
}

void setupEspNow() {
    WiFi.mode(WIFI_STA);
    Serial.print("[ESP-NOW] Sender MAC: ");
    Serial.println(WiFi.macAddress());

    if (esp_now_init() != ESP_OK) {
        Serial.println("[ESP-NOW] init failed. restart.");
        ESP.restart();
    }

    esp_now_register_send_cb(onSent);

    esp_now_peer_info_t peer = {};
    memcpy(peer.peer_addr, receiverMAC, 6);
    peer.channel = 0;
    peer.encrypt = false;

    if (esp_now_add_peer(&peer) == ESP_OK) Serial.println("[ESP-NOW] receiver peer added.");
    else Serial.println("[ESP-NOW] peer add failed. Check receiverMAC[].");
}

void readGps() {
    while (gpsSerial.available() > 0) gps.encode(gpsSerial.read());

    if (gps.location.isUpdated()) {
        lastLat = gps.location.lat();
        lastLng = gps.location.lng();
        lastGpsValid = gps.location.isValid() ? 1 : 0;
    }
    if (gps.speed.isUpdated()) lastSpeed = gps.speed.mps();
    if (gps.course.isUpdated()) lastHeading = gps.course.deg();
}

void readImuIntoPacket() {
    if (imu.dataReady()) {
        imu.getAGMT();
        packet.accel_x = (imu.accX() / 1000.0) * 9.80665;
        packet.accel_y = (imu.accY() / 1000.0) * 9.80665;
        packet.accel_z = (imu.accZ() / 1000.0) * 9.80665;
        packet.gyro_x = imu.gyrX();
        packet.gyro_y = imu.gyrY();
        packet.gyro_z = imu.gyrZ();
    }
}

uint8_t decideRiskForTest(float speed, float ax, float ay, float az) {
    float totalA = sqrt(ax * ax + ay * ay + az * az);
    float motionA = fabs(totalA - 9.80665);

    if (speed > 2.0 || motionA > 5.0) return RISK_WARNING;
    if (speed > 1.0 || motionA > 2.5) return RISK_CAUTION;
    return RISK_SAFE;
}

void fillAndSendPacket() {
    packet.node_type = NODE_CANE;
    packet.gps_valid = lastGpsValid;
    packet.latitude = lastLat;
    packet.longitude = lastLng;
    packet.speed_mps = lastSpeed;
    packet.heading_deg = lastHeading;
    packet.risk_level = decideRiskForTest(packet.speed_mps, packet.accel_x, packet.accel_y, packet.accel_z);
    packet.timestamp_ms = millis();
    packet.seq_num = seq++;

    esp_err_t result = esp_now_send(receiverMAC, (uint8_t *)&packet, sizeof(packet));
    if (result != ESP_OK) Serial.println("[ESP-NOW] send function error.");

    Serial.printf("[SEND] seq=%u gps=%u lat=%.6f lng=%.6f spd=%.2f ax=%.2f ay=%.2f az=%.2f risk=%u size=%d\n",
        packet.seq_num, packet.gps_valid, packet.latitude, packet.longitude,
        packet.speed_mps, packet.accel_x, packet.accel_y, packet.accel_z,
        packet.risk_level, sizeof(packet));
}

void setup() {
    Serial.begin(115200);

```

```

delay(1000);
Serial.println("=== V2X Smart Cane Sender: GPS + IMU + ESP-NOW ===");
memset(&packet, 0, sizeof(packet));
setupImu();
setupGps();
setupEspNow();
}

void loop() {
  readGps();
  readImuIntoPacket();

  if (millis() - lastSendMs >= SEND_PERIOD_MS) {
    lastSendMs = millis();
    fillAndSendPacket();
  }
}

```

A-8. 송신기 정상 출력 확인 95~110분

```

[SEND] seq=0 gps=0 lat=0.000000 lng=0.000000 spd=0.00 ax=0.12 ay=-0.05 az=9.78 risk=0 size=55
[ESP-NOW TX] seq=0 result=OK
[SEND] seq=1 gps=1 lat=37.496123 lng=126.957456 spd=0.14 ax=0.20 ay=-0.02 az=9.81 risk=0 size=55

```

- **seq**가 0, 1, 2, 3으로 증가하면 10Hz 송신 루프가 정상이다.
- **ax/ay/az**가 보드를 기울일 때 변하면 IMU 매핑이 정상이다.
- **gps=1**과 lat/lng가 0이 아닌 값이면 GPS 좌표 수신이 정상이다.
- **result=FAIL**이 반복되면 receiverMAC[]가 틀렸을 가능성이 가장 높다.

A-9. 박종선 파트 완료 기준

체크	완료 기준	증거
☐	I2C 스캐너에서 0x69 또는 0x68 검출	시리얼 캡처
☐	GPS charsProcessed 증가 및 야외 좌표 수신	시리얼 캡처
☐	Sender 코드에서 [ESP-NOW TX] result=OK 반복	시리얼 캡처
☐	보드 움직임에 따라 accel_x/y/z 값 변화	짧은 영상

파트 B. 강현준 — 수신기 ESP-NOW + Jetson Nano UART 파이프라인

오늘의 목표: 박중선의 송신기에서 오는 실제 센서 패킷을 수신하고, 같은 데이터를 Jetson Nano로 지연 없이 넘겨 AI 연산 입력 데이터 형태를 만든다.

B-1. 준비물 체크 0~10분

체크	준비물	개수	확인 방법
☐	ESP32 DevKit 수신기	1개	보드에 “수신기” 라벨 부착
☐	Jetson Nano	1대	부팅 및 터미널 접근 확인
☐	점퍼선 암-수	3개 이상	Jetson 40핀 헤더 연결용
☐	USB 케이블	1개	ESP32 업로드용
☐	Python pyserial	1개	pip 설치 확인

B-2. 수신기 MAC 주소 확인 10~20분

```
#include <WiFi.h>

void setup() {
  Serial.begin(115200);
  delay(1000);
  WiFi.mode(WIFI_STA);
  Serial.println("=====");
  Serial.print("Receiver ESP32 MAC: ");
  Serial.println(WiFi.macAddress());
  Serial.println("=====");
}

void loop() {}
```

출력 예시: Receiver ESP32 MAC: C8:F0:9E:52:78:A4. 이 값을 박중선에게 전달하여 Sender 코드 receiverMAC[]에 넣게 한다.

B-3. Jetson Nano UART 배선 20~35분

ESP32 수신기	Jetson Nano 40핀 헤더	역할	주의
GPIO17 TX2	Pin 10 UART RXD	ESP32가 Jetson으로 데이터 전송	TX → RX 교차
GPIO16 RX2	Pin 8 UART TXD	Jetson이 ESP32로 명령 전송	초기에는 없어도 되지만 연결 권장
GND	Pin 6 또는 Pin 9 GND	공통 접지	GND 없으면 문자가 깨짐
5V/3V3	연결하지 않음	전원 분리	신호선과 GND만 연결

금지: Jetson의 5V 핀을 ESP32 3V3 핀에 연결하지 않는다. 두 보드는 각자 전원을 사용하고 GND만 공유한다.

B-4. 수신기 최종 코드 업로드 35~70분

아래 코드는 ESP-NOW 수신, Jetson UART 전달, 액추에이터 제어 호출까지 포함한다. 액추에이터가 아직 연결되지 않았으면 코드 상단의 **USE_ACTUATOR**를 0으로 바꿔 먼저 수신만 테스트한다.


```

#include <WiFi.h>
#include <esp_now.h>
#include <DFRobotDFPlayerMini.h>

#define NODE_CANE 0x01
#define RISK_SAFE 0
#define RISK_CAUTION 1
#define RISK_WARNING 2
#define RISK_DANGER 3

#define LED_PIN 2
#define JETSON_RX 16
#define JETSON_TX 17
#define JETSON_BAUD 115200
#define DF_RX 32
#define DF_TX 33
#define MOTOR_PWM 25
#define MOTOR_ENABLE 26
#define PWM_CH 0
#define PWM_FREQ 5000
#define PWM_RES 8

#define USE_ACTUATOR 1
#define USE_JETSON_UART 1

typedef struct __attribute__((packed)) v2x_message {
    uint8_t node_type;
    uint8_t risk_level;
    uint8_t gps_valid;
    float latitude;
    float longitude;
    float speed_mps;
    float heading_deg;
    float accel_x;
    float accel_y;
    float accel_z;
    float gyro_x;
    float gyro_y;
    float gyro_z;
    uint32_t timestamp_ms;
    uint16_t seq_num;
} v2x_message_t;

HardwareSerial jetsonSerial(2);
HardwareSerial dfSerial(1);
DFRobotDFPlayerMini player;
v2x_message_t rx;
uint32_t recvCount = 0;
uint32_t lostCount = 0;
uint16_t prevSeq = 0;
uint32_t lastRxMs = 0;
uint8_t currentRisk = 255;
uint32_t lastVoiceMs = 0;

void setupActuator() {
    #if USE_ACTUATOR
        pinMode(MOTOR_ENABLE, OUTPUT);
        digitalWrite(MOTOR_ENABLE, HIGH);
        ledcSetup(PWM_CH, PWM_FREQ, PWM_RES);
        ledcAttachPin(MOTOR_PWM, PWM_CH);
        ledcWrite(PWM_CH, 0);

        dfSerial.begin(9600, SERIAL_8N1, DF_RX, DF_TX);
        if (player.begin(dfSerial)) {
            player.volume(24);
            Serial.println("[DFPlayer] OK. SD files: /mp3/0001.mp3 ~ 0004.mp3");
        } else {
            Serial.println("[DFPlayer] FAIL. Continue without voice.");
        }
    #endif
}

void applyRisk(uint8_t risk) {
    #if USE_ACTUATOR
        uint8_t duty = 0;
        uint8_t track = 0;

        if (risk == RISK_SAFE) {
            duty = 0;
            track = 1;
        } else if (risk == RISK_CAUTION) {
            duty = 80;
            track = 2;
        } else if (risk == RISK_WARNING) {
            duty = 160;
            track = 3;
        } else {
            duty = 255;
            track = 4;
        }

        ledcWrite(PWM_CH, duty);

        if (risk != currentRisk || millis() - lastVoiceMs > 3000) {

```

```

        currentRisk = risk;
        lastVoiceMs = millis();
        if (track > 0) player.play(track);
    }
#endif
}

void sendJsonToJetson(const v2x_message_t &m) {
#ifdef USE_JETSON_UART
    jetsonSerial.printf(
        "{ \"seq\":%u, \"node\":%u, \"risk\":%u, \"gps_valid\":%u, \"lat\":%.6f, \"lng\":%.6f, \"speed\":%.3f, \"heading\":%.2f, \"ax\":%.3f, \"ay\":%.3f, \"az\":%.3f, \"gx\":%.3f, \"gy\":%.3f, \"gz\":%.3f, \"tx_ms\":%lu, \"rx_ms\":%lu} \n",
        m.seq_num, m.node_type, m.risk_level, m.gps_valid,
        m.latitude, m.longitude, m.speed_mps, m.heading_deg,
        m.accel_x, m.accel_y, m.accel_z,
        m.gyro_x, m.gyro_y, m.gyro_z,
        (unsigned long)m.timestamp_ms, (unsigned long)millis()
    );
#endif
}

void onDataRecv(const esp_now_recv_info_t *info, const uint8_t *data, int len) {
    if (len != sizeof(v2x_message_t)) {
        Serial.printf("[RX] size mismatch. len=%d expected=%d\n", len, sizeof(v2x_message_t));
        return;
    }

    memcpy(&rx, data, sizeof(rx));
    recvCount++;

    if (recvCount > 1) {
        uint16_t expected = prevSeq + 1;
        if (rx.seq_num != expected) lostCount += (uint16_t)(rx.seq_num - expected);
    }
    prevSeq = rx.seq_num;
    lastRxMs = millis();

    digitalWrite(LED_PIN, HIGH);

    Serial.printf("[RX] seq=%u risk=%u gps=%u lat=%.6f lng=%.6f spd=%.2f ax=%.2f ay=%.2f az=%.2f recv=%lu lost=%lu\n",
        rx.seq_num, rx.risk_level, rx.gps_valid,
        rx.latitude, rx.longitude, rx.speed_mps,
        rx.accel_x, rx.accel_y, rx.accel_z,
        (unsigned long)recvCount, (unsigned long)lostCount);

    sendJsonToJetson(rx);
    applyRisk(rx.risk_level);

    digitalWrite(LED_PIN, LOW);
}

void setupEspNowReceiver() {
    WiFi.mode(WIFI_STA);
    Serial.print("[ESP-NOW] Receiver MAC: ");
    Serial.println(WiFi.macAddress());

    if (esp_now_init() != ESP_OK) {
        Serial.println("[ESP-NOW] init failed. restart.");
        ESP.restart();
    }
    esp_now_register_recv_cb(onDataRecv);
}

void setup() {
    Serial.begin(115200);
    delay(1000);
    pinMode(LED_PIN, OUTPUT);

#ifdef USE_JETSON_UART
    jetsonSerial.begin(JETSON_BAUD, SERIAL_8N1, JETSON_RX, JETSON_TX);
    Serial.println("[Jetson UART] ESP32 TX17 -> Jetson RX, ESP32 RX16 <- Jetson TX");
#endif

    setupActuator();
    setupEspNowReceiver();
    Serial.println("=== V2X Receiver: ESP-NOW + Jetson UART + Actuator ===");
}

void loop() {
    if (millis() - lastRxMs > 1000) {
        digitalWrite(LED_PIN, (millis() / 250) % 2);
    }
    delay(10);
}

```

Arduino ESP32 Core 버전 주의: 최신 Core 3.x는 onDataRecv 인자가 형태다. 만약 컴파일 오류가 나면 ESP32 보드 패키지를 최신으로 업데이트한다.

```
const esp_now_recv_info_t *info
```

B-5. Jetson Nano UART 설정 70~90분

```
sudo usermod -aG dialout $USER
sudo systemctl stop nvgetty
sudo systemctl disable nvgetty
sudo reboot
ls -l /dev/ttyTHS1
python3 -m pip install pyserial pandas
```

- **/dev/ttyTHS1**가 기본 UART 포트인 경우가 많다.
- 데이터가 안 들어오면 Python 코드에서 PORT를 **/dev/ttyS0**로 바꿔 테스트한다.
- nvgetty가 UART를 점유하면 Python이 데이터를 못 받을 수 있으므로 비활성화한다.

B-6. Jetson Python 수신 코드 실행 90~115분

```
import json
import time
import serial
from datetime import datetime

PORT = "/dev/ttyTHS1" # If no data, try "/dev/ttyS0"
BAUD = 115200
LOG_PATH = "v2x_uart_log.csv"

header = "pc_time,seq,node,risk,gps_valid,lat,lng,speed,heading,ax,ay,az,gx,gy,gz,tx_ms,rx_ms\n"

ser = serial.Serial(PORT, BAUD, timeout=0, write_timeout=0)
print(f"[Jetson] listening {PORT} @ {BAUD}")

buf = ""
last_seq = None

with open(LOG_PATH, "w", encoding="utf-8") as f:
    f.write(header)

    while True:
        data = ser.read(4096)
        if not data:
            time.sleep(0.002)
            continue

        buf += data.decode("utf-8", errors="ignore")

        while "\n" in buf:
            line, buf = buf.split("\n", 1)
            line = line.strip()
            if not line:
                continue

            try:
                d = json.loads(line)
            except json.JSONDecodeError:
                print("[JSON FAIL]", line[:120])
                continue

            seq = int(d["seq"])
            if last_seq is not None and seq != ((last_seq + 1) & 0xFFFF):
                print(f"[PACKET GAP] prev={last_seq}, now={seq}")
            last_seq = seq

            pc_time = datetime.now().isoformat(timespec="milliseconds")
            row = [pc_time, d["seq"], d["node"], d["risk"], d["gps_valid"],
                  d["lat"], d["lng"], d["speed"], d["heading"],
                  d["ax"], d["ay"], d["az"], d["gx"], d["gy"], d["gz"],
                  d["tx_ms"], d["rx_ms"]]
            f.write(",".join(map(str, row)) + "\n")
            f.flush()

            print(f"seq={seq:05d} risk={d['risk']} gps={d['gps_valid']} "
                  f"lat={d['lat']:.6f} lng={d['lng']:.6f} "
                  f"speed={d['speed']:.2f} ax={d['ax']:.2f} ay={d['ay']:.2f} az={d['az']:.2f}")
```

B-7. 강현준 정상 출력 확인

```
[Jetson] listening /dev/ttyTHS1 @ 115200
seq=000001 risk=0 gps=1 lat=37.496123 lng=126.957456 speed=0.14 ax=0.20 ay=-0.02 az=9.81
seq=000002 risk=1 gps=1 lat=37.496126 lng=126.957460 speed=0.18 ax=2.92 ay=-0.11 az=9.33
```

증상	원인	해결
----	----	----

Jetson에 아무것도 안 뜸	TX/RX 반대, GND 미연결, 포트명 오류	GPIO17→Jetson RX, GND 공통, /dev/ttyTHS1↔/dev/ttyS0 변경
문자가 깨짐	Baudrate 불일치	ESP32와 Python 모두 115200 확인
JSON FAIL 반복	Serial 출력과 UART 출력 혼동, 줄 중간 끊김	jetsonSerial에는 JSON만 출력, 개행 확인
PACKET GAP 발생	ESP-NOW 유실 또는 UART 처리 지연	송신 주기 100ms 유지, Serial 출력 줄이기

B-8. 강현준 파트 완료 기준

체크	완료 기준	증거
☐	수신기 MAC 주소를 박중선에게 전달	MAC 캡처
☐	수신기 Serial Monitor에서 실제 센서 패킷 출력	수신 로그
☐	Jetson 터미널에서 JSON 파싱 결과 출력	터미널 캡처
☐	v2x_uart_log.csv 파일 저장	CSV 파일

파트 C. 박종선 — 수신기 측 진동/음성 액추에이터 회로

오늘의 목표: 수신기 ESP32가 받은 risk_level에 따라 원형 진동 모터의 세기와 DFPlayer Mini 음성 트랙을 다르게 출력한다.

C-1. 준비물 체크

체크	준비물	개수	비고
☐	DFPlayer Mini	1개	microSD 카드 필요
☐	8Ω 스피커	1개	1~3W 권장
☐	원형 진동 모터	1개 이상	정격 전압 확인
☐	NPN 트랜지스터 또는 N-MOSFET	1개	Low-side 스위칭
☐	1kΩ 저항	2개	DFPlayer RX 직렬, 모터 베이스/게이트
☐	플라이백 다이오드	1개	모터 역기전력 보호
☐	외부 5V 전원	1개	모터/DFPlayer 안정 전원

C-2. DFPlayer Mini 배선

DFPlayer 핀	ESP32/부품 연결	역할	주의
VCC	외부 5V	음성 모듈 전원	전류 부족 시 재생 실패
GND	ESP32 GND와 공통	접지	공통 GND 필수
TX	ESP32 GPIO32 RX1	DFPlayer 응답	교차 연결
RX	ESP32 GPIO33 TX1 → 1kΩ → DF RX	ESP32 명령 수신	1kΩ 직렬 저항 권장
SPK_1, SPK_2	스피커 양단	오디오 출력	스피커 직접 연결

microSD 파일명: /mp3/0001.mp3 = 안전합니다, /mp3/0002.mp3 = 주의하세요, /mp3/0003.mp3 = 차량이 접근합니다, /mp3/0004.mp3 = 위험합니다 즉시 정지하세요.

C-3. 진동 모터 Low-side 스위칭 배선

연결 지점	연결 대상	설명
모터 +	외부 5V +	모터 전원은 ESP32 GPIO에서 공급하지 않음
모터 -	NPN 컬렉터 또는 MOSFET 드레인	스위칭되는 쪽
NPN 이미터 또는 MOSFET 소스	GND	Low-side 스위치
ESP32 GPIO25	1kΩ → 베이스/게이트	PWM 제어
다이오드	모터 양단 역방향 병렬	역기전력으로 ESP32 보호

중요: 원형 진동 모터를 ESP32 GPIO에 직접 연결하면 보드가 손상될 수 있다. 반드시 트랜지스터 또는 MOSFET을 사용한다.

C-4. 액추에이터 단독 테스트 코드

```
#include <DFRobotDFPlayerMini.h>

#define DF_RX 32
#define DF_TX 33
#define MOTOR_PWM 25
#define MOTOR_ENABLE 26
#define PWM_CH 0
#define PWM_FREQ 5000
#define PWM_RES 8

HardwareSerial dfSerial(1);
DFRobotDFPlayerMini player;

void setMotor(uint8_t duty) {
    ledcWrite(PWM_CH, duty);
}

void setup() {
    Serial.begin(115200);
    pinMode(MOTOR_ENABLE, OUTPUT);
    digitalWrite(MOTOR_ENABLE, HIGH);

    ledcSetup(PWM_CH, PWM_FREQ, PWM_RES);
    ledcAttachPin(MOTOR_PWM, PWM_CH);
    setMotor(0);

    dfSerial.begin(9600, SERIAL_8N1, DF_RX, DF_TX);
    if (player.begin(dfSerial)) {
        player.volume(24);
        Serial.println("DFPlayer OK");
    } else {
        Serial.println("DFPlayer FAIL. Check SD, power, TX/RX.");
    }

    Serial.println("Input 0,1,2,3 in Serial Monitor.");
}

void loop() {
    if (!Serial.available()) return;
    char c = Serial.read();

    if (c == '0') {
        setMotor(0);
        player.play(1);
        Serial.println("risk 0: safe");
    } else if (c == '1') {
        setMotor(80);
        player.play(2);
        Serial.println("risk 1: caution");
    } else if (c == '2') {
        setMotor(160);
        player.play(3);
        Serial.println("risk 2: warning");
    } else if (c == '3') {
        setMotor(255);
        player.play(4);
        Serial.println("risk 3: danger");
    }
}
```

C-5. 수신기 코드와 통합하는 방법

1. C-4 단독 테스트에서 Serial Monitor에 0, 1, 2, 3을 입력해 진동 세기와 음성 트랙이 바뀌는지 확인한다.
2. 정상 동작하면 B-4의 수신기 최종 코드에서 **USE_ACTUATOR 1** 상태로 둔다.
3. onDataRecv() 내부에서 이미 **applyRisk(rx.risk_level)**이 호출되므로, 수신 패킷의 risk_level이 바뀌면 자동으로 출력이 바뀐다.
4. Sender 코드의 decideRiskForTest() 임계값을 낮추거나 임시로 packet.risk_level = 1, 2, 3을 넣어 출력 테스트를 한다.

C-6. 액추에이터 완료 기준

체크	완료 기준	증거
----	-------	----

☐	DFPlayer에서 0001~0004.mp3가 각각 재생됨	영상
☐	risk 1, 2, 3에 따라 진동 세기가 달라 짐	영상
☐	모터 작동 시 ESP32가 리셋되지 않음	연속 테스트
☐	수신기 ESP-NOW 패킷과 액추에이터 가 동시에 동작	통합 영상

파트 D. 최민서 — SUMO 차량 Flow 주입 및 AI 학습용 FCD 추출

오늘의 목표: Import한 주차장 진출입로 맵에 20대 이상의 차량을 투입하고, 각 차량의 시간별 위치·속도·방향각 데이터를 CSV로 변환한다.

D-1. 준비물 및 파일 체크 0~10분

체크	항목	확인 방법
☐	SUMO 설치	터미널에서 <code>sumo --version</code>
☐	netedit 실행	터미널에서 <code>netedit</code>
☐	주차장 진출입로 net 파일	<code>parking_lot.net.xml</code> 존재
☐	작업 폴더	<code>V2X_SUMO_sim</code> 폴더 사용

D-2. Edge ID 확인 10~25분

- netedit에서 `parking_lot.net.xml`을 연다.
- 키보드 `I`를 눌러 Inspect 모드로 전환한다.
- 차량이 출발할 도로, 교차로, 도착 도로를 클릭하여 Edge ID를 적는다.
- 반대 방향 도로는 SUMO에서 일반적으로 -E0처럼 마이너스 기호가 붙는다.

역할	내 맵 Edge ID 기록	예시
주차장 출구 도로		E_parking
보도/교차 지점 도로		E_cross
메인 도로		E_main
회전/측면 도로		E_side

D-3. parking_lot.rou.xml 작성 25~55분

```
<routes>
  <vType id="car_normal" vClass="passenger" color="blue" accel="2.0" decel="4.5" length="5.0"
maxSpeed="8.33"/>
  <vType id="car_fast" vClass="passenger" color="red" accel="3.0" decel="5.0" length="5.0"
maxSpeed="11.11"/>
  <vType id="car_slow" vClass="passenger" color="green" accel="1.2" decel="3.5" length="5.0"
maxSpeed="5.56"/>

  <!-- 아래 Edge ID는 예시다. netedit Inspect에서 본인 맵의 Edge ID로 반드시 수정한다. -->
  <route id="route_exit" edges="E_parking E_cross E_main"/>
  <route id="route_enter" edges="-E_main -E_cross -E_parking"/>
  <route id="route_turn" edges="E_parking E_cross E_side"/>

  <flow id="flow_exit_normal" type="car_normal" route="route_exit" begin="0" end="120" number="10"
departSpeed="max"/>
  <flow id="flow_exit_fast" type="car_fast" route="route_exit" begin="20" end="120" number="6"
departSpeed="max"/>
  <flow id="flow_enter_slow" type="car_slow" route="route_enter" begin="10" end="120" number="8"
departSpeed="max"/>
  <flow id="flow_turn_fast" type="car_fast" route="route_turn" begin="40" end="130" number="6"
departSpeed="max"/>
</routes>
```


반드시 수정: E_parking, E_cross, E_main, E_side는 예시 ID다. D-2에서 직접 확인한 실제 Edge ID로 바꾸지 않으면 차량이 생성되지 않는다.

D-4. parking_lot.sumocfg 작성 55~65분

```
<configuration>
  <input>
    <net-file value="parking_lot.net.xml"/>
    <route-files value="parking_lot.rou.xml"/>
  </input>
  <time>
    <begin value="0"/>
    <end value="200"/>
    <step-length value="0.1"/>
  </time>
  <output>
    <fcd-output value="fcd_output.xml"/>
    <fcd-output.geo value="false"/>
  </output>
</configuration>
```

D-5. SUMO 실행 65~90분

```
cd V2X_SUMO_sim
sumo-gui -c parking_lot.sumocfg
```

- GUI가 열리면 재생 버튼을 누른다.
- 차량이 너무 빠르면 Delay를 100ms로 설정한다.
- 차량이 안 나오면 rou.xml의 Edge ID와 route 방향을 다시 확인한다.

```
sumo -c parking_lot.sumocfg
```

CLI 실행이 끝나면 같은 폴더에 **fcd_output.xml**이 생성되어야 한다.

D-6. FCD XML → CSV 전처리 90~120분

```
import xml.etree.ElementTree as ET
from pathlib import Path
import pandas as pd

src = Path("fcd_output.xml")
out = Path("fcd_trajectory.csv")
rows = []

for event, elem in ET.iterparse(src, events=("end",)):
    if elem.tag == "timestep":
        t = float(elem.attrib["time"])
        for v in elem.findall("vehicle"):
            rows.append({
                "time": t,
                "vehicle_id": v.attrib.get("id", ""),
                "x": float(v.attrib.get("x", 0.0)),
                "y": float(v.attrib.get("y", 0.0)),
                "speed": float(v.attrib.get("speed", 0.0)),
                "angle": float(v.attrib.get("angle", 0.0)),
                "lane": v.attrib.get("lane", ""),
                "pos": float(v.attrib.get("pos", 0.0)),
            })
        elem.clear()

df = pd.DataFrame(rows)
if df.empty:
    raise RuntimeError("No vehicle data. Check route file and SUMO simulation.")

df = df.sort_values(["vehicle_id", "time"]).reset_index(drop=True)
df["dt"] = df.groupby("vehicle_id")["time"].diff()
df["dv"] = df.groupby("vehicle_id")["speed"].diff()
df["accel_est"] = df["dv"] / df["dt"]
df["dx"] = df.groupby("vehicle_id")["x"].diff()
df["dy"] = df.groupby("vehicle_id")["y"].diff()
df["is_anomaly_candidate"] = (df["accel_est"].abs() > 3.0) | (df["speed"] > 10.0)

df.to_csv(out, index=False, encoding="utf-8-sig")
print(df.head(30))
print("vehicle count:", df["vehicle_id"].nunique())
```

```
print("row count:", len(df))
print("saved:", out)
```

D-7. 최민서 정상 출력 확인

vehicle count: 30
row count: 15420
saved: fcd_trajectory.csv

열 이름	의미	AI 활용
time	시뮬레이션 시간	시계열 입력 시간축
vehicle_id	차량 ID	차량별 궤적 분리
x, y	차량 좌표	Transformer 위치 입력
speed	속도	위험도, 급가속 탐지
angle	진행 방향각	회전/진입 패턴 분석
accel_est	속도 차분 기반 추정 가속도	이상 주행 후보 라벨

D-8. 최민서 파트 완료 기준

체크	완료 기준	증거
☐	sumo-gui에서 차량 20대 이상 이동	화면 캡처
☐	parking_lot.rou.xml과 parking_lot.sumocfg 저장	파일
☐	fcd_output.xml 생성	파일
☐	fcd_trajectory.csv 생성 및 vehicle count 출력	터미널 캡처

파트 E. 전원 — 3.5시간 이후 End-to-End 통합 디버깅

목표: 송신기 센서 → ESP-NOW → 수신기 → Jetson UART → 액추에이터까지 한 번에 이어지는지 검증한다.

E-1. 통합 순서

- 수신기 먼저 실행:** 강현준이 수신기 ESP32를 연결하고 Serial Monitor에 Receiver MAC과 수신 대기 상태를 띄운다.
- Jetson Python 실행:** 강현준이 Jetson에서 UART 파서를 실행하고 “listening” 상태를 확인한다.
- 송신기 실행:** 박중선이 송신기 ESP32를 켜고 [SEND], [ESP-NOW TX] OK를 확인한다.
- 수신 확인:** 강현준이 수신기 Serial Monitor에서 [RX] seq 증가와 lostCount를 확인한다.
- Jetson 확인:** Jetson 터미널에서 seq, risk, gps_valid, lat, lng, ax, ay, az가 출력되는지 확인한다.
- 액추에이터 확인:** 박중선이 risk_level을 1, 2, 3으로 바꿔 진동·음성 출력이 바뀌는지 확인한다.
- 야외 GPS 테스트:** 전원이 안정적이면 야외로 이동하여 GPS cold start 후 좌표가 변하는지 확인한다.

E-2. 통합 테스트 기록 양식

테스트	합격 기준	결과 기록
ESP-NOW 1분 연속 수신	seq 연속, lost 5% 이내	
Jetson UART 1분 연속 수신	JSON FAIL 없음 또는 매우 적음	
IMU 움직임 반응	보드 흔들 때 ax/ay/az 즉시 변화	
GPS 야외 수신	gps_valid=1, lat/lng 정상	
risk 1 출력	약한 진동 + 0002.mp3	
risk 2 출력	중간 진동 + 0003.mp3	
risk 3 출력	강한 진동 + 0004.mp3	

E-3. 4차 미팅 종료 체크리스트

체크	완료 조건	담당
☐	송신기 GPS+IMU 실제 데이터가 10Hz로 송신됨	박중선
☐	수신기에서 ESP-NOW 패킷과 유실률이 출력됨	강현준
☐	Jetson Nano에서 UART JSON 데이터가 실시간 파싱됨	강현준
☐	risk_level별 진동/음성 출력이 동작함	박중선
☐	SUMO 차량 Flow와 FCD CSV가 생성됨	최민서
☐	팀 채팅방에 배선 사진, 로그 캡처, SUMO 캡처를 업로드함	전원

부록 1. 5차·6차 하드웨어 프로토타입 제작 주의사항

1. 전원 분리

부품군	권장 전원	이유
Jetson Nano	5V 4A 이상 독립 전원	AI 연산 순간 전류로 ESP32 브라운아웃 방지
ESP32 + GPS + IMU	안정된 5V USB 또는 3.3V 레귤레이터	센서/통신 안정성 확보
진동 모터 + DFPlayer	별도 5V 라인 권장	모터 기동 전류와 오디오 피크 전류 대응

2. 납땜 전환

- 5차부터 브레드보드를 줄이고 만능기판 + 실리콘 와이어 + 수축튜브로 전환한다.
- GPS, IMU, DFPlayer, 모터 선은 JST 커넥터를 사용하면 분해와 수리가 쉽다.
- 지팡이는 충격이 많으므로 납땜부를 글루건으로 고정한다.

3. 폼팩터

- 지팡이에는 ESP32, GPS, IMU, 진동 모터, 소형 스피커 정도만 올린다.
- Jetson Nano와 대용량 배터리는 백팩 또는 크로스백에 넣는 구조가 현실적이다.
- 지팡이-가방 사이에는 UART/USB 케이블을 사용하고, 케이블 당김 방지 클립을 단다.

부록 2. 빠른 문제 해결표

문제	가장 흔한 원인	해결
IMU 주소 안 뜸	SDA/SCL 반대, VCC/GND 불량	I2C 스캐너로 0x69 확인, ADO_VAL 0/1 변경
GPS 좌표 안 뜸	실내, TX/RX 오류	야외 3분 대기, GPS TX→ESP32 RX16 확인
ESP-NOW FAIL	MAC 주소 오타	수신기 MAC 재확인 후 receiverMAC[] 수정
수신 데이터 깨짐	구조체 불일치	v2x_message_t를 송수신기 완전 동일 복사
Jetson 수신 안 됨	TX/RX 반대, GND 없음, 포트명 다름	배선 교차, 공통 GND, /dev/ttyTHS1 또는 /dev/ttyS0 확인
모터 돌 때 리셋	브라운아웃	모터 전원 분리, 다이오드, 커패시터 추가
DFPlayer 무음	SD 파일명/전원/TX-RX 문제	/mp3/0001.mp3 형식, 5V 공급, 1kΩ 저항 확인
SUMO 차량 안 나옴	Edge ID 오류	netedit Inspect로 실제 Edge ID 반영